

PL/SQL H1

Introductie

HOGESCHOOL 



Wat is PL/SQL?

- PL/SQL
- Procedural Language/Structured Query Language
- Maakt het mogelijk om SQL (een niet-procedurele 4^{de} generatietaal) uit te breiden met elementen uit een procedurele 3^{de} generatietaal

Mogelijkheden SQL

- SQL: nodig om relationele database te maken en te beheren
- 4GL = declaratief



definiëren **wat** we willen, niet **hoe** we tot een resultaat komen

vb. `SELECT * FROM medewerkers;`

Onmogelijkheden SQL

- gebruik van variabelen
(subquery = gedeeltelijke opl.)
- records in bepaalde volgorde verwerken
- ontbreken procedurele constructies
 - voorwaarde
 - iteratie

Mogelijkheden PL/SQL

- gebruik variabelen en constanten
- programmabesturing (iteratie en selectie)
- mogelijkheden om rij per rij-verwerking te kunnen doen
- modulariteit: programma's opdelen in kleinere eenheden
- afschermen van broncode door packages
- foutafhandeling (exceptions)


PL/SQL Programming Environments

- SQL Developer
- SQL*Plus (idem opleidingsonderdeel Data)
- Jdeveloper

SQL*Plus

- Vooraf Oracle XE installeren volgens installatieprocedure op BB
- Gebruikte tabellen: idem opleidingsonderdeel Data

The Human Resources (HR) Schema



PL/SQL H2

Functies

HOGESCHOOL 



Ingebouwde functies

- Bekend vanuit de lessen SQL

Vb SELECT sysdate FROM dual;
 SELECT SUBSTR(last_name, 1, 4)
 FROM employees;

Er bestaan functies met of zonder parameters.
Een functie geeft altijd 1 resultaat terug.

Wat is een functie?

- object (bestaat uit statements en PL/SQL-constructies) met een naam
- wordt bewaard in de DB
- code op 1 plaats definiëren en op meerdere plaatsen gebruiken
- retourneert een waarde

Syntax voor de creatie van een functie

Het PL/SQL blok moet minstens 1 RETURN statement bevatten.

```
CREATE [OR REPLACE] FUNCTION function_name
  [(parameter1 [mode1] datatype1, ...)]
RETURN datatype IS|AS
  [local_variable_declarations; ...]
BEGIN
  -- actions;
  RETURN expression;
END [function_name];
```

Voorbeeld: functie zonder parameters

```
CREATE OR REPLACE FUNCTION sysdate2
```

```
RETURN VARCHAR2
```

Return-type: geen lengte!!

```
AS
```

```
    v_tekst    VARCHAR2(30);
```

Declaratie van variabelen

```
BEGIN
```

Statements

```
    v_tekst := TO_CHAR(sysdate, 'fmDay, dd month yyyy');
```

```
    RETURN v_tekst;
```

```
END;
```

```
/
```

PL/SQL Syntax

- Elke instructie eindigt met ;
- Een toekenning gebeurt door :=
- In een PL/SQL blok kan je andere functies oproepen, je hoeft hiervoor geen SELECT-statement te gebruiken (behalve bij groepsfuncties en DECODE).

Vb. v_lengte := LENGTH(v_naam);
 v_rest := MOD(v_lengte, 2);
 v_hoofd := UPPER(v_naam);

- Commentaar toevoegen aan je code:
 - 1 regel: -- dit is 1 lijn commentaar
 - meerdere regels: /* commentaar over meerdere lijnen */

Opmerkingen bij functies

- Een functie heeft (in tegenstelling tot een gewoon PL/SQL blok) altijd een naam. Een gewoon PL/SQL blok noemt men ook wel **een ANONIEM blok**.
- Bij het RETURN-type mag geen lengte meegegeven worden.
- Tussen IS/AS en BEGIN kan je variabelen declareren (zie verder).
- In de body van de functie → minstens 1 return-commando met daarachter de waarde die wordt teruggegeven (van het type zoals hoger beschreven!)
- Functies kunnen van een **argumentenlijst** worden voorzien, door deze argumenten of **parameters** wordt de flexibiliteit vergroot (zie verder).

/ → creatie functie

- de broncode wordt altijd in de data dictionary opgeslagen
- als foutloze code: gecompileerde versie → databank
- als code met fouten:
Melding: ``created with compilation errors'`.`

Hoe fouten opvragen: `show errors`

Functie gebruiken

- Vanuit een andere functie:

in het BEGIN-blok: `v_datum := sysdate2;`

- Vanuit een SQL-instructie:

SQL> `SELECT sysdate2 FROM dual;`

Declaratie en initialisatie van PL/SQL variabelen

Syntax:

```
identifier [CONSTANT] datatype [NOT NULL]
           [:= | DEFAULT expr];
```

Voorbeelden:

```
v_hiredate    DATE;
v_deptno      NUMBER(2) NOT NULL := 10;
v_location    VARCHAR2(13) := 'Atlanta';
v_comm        CONSTANT NUMBER := 1400;
```

Naamgeving variabelen

- Moet beginnen met een letter
 - Mag letters en getallen bevatten
 - Mag bevatten: dollar teken (\$), underscore, pond teken (£)
 - Maximale lengte is 30 tekens
 - Geen gereserveerde woorden
-
- Opmerking: naam begint met v_

Belangrijkste Scalar Data Types

- CHAR [(maximum_lengte)]
- VARCHAR2 (maximum_lengte)
- LONG
- NUMBER [(precisie, schaal)]
- BINARY_INTEGER
- PLS_INTEGER
- BOOLEAN
- BINARY_FLOAT: sneller, maar minder precies
- BINARY_DOUBLE
- DATE
- TIMESTAMP

Declaratie: %TYPE

verwijzen naar het datatype van een andere variabele

- datatype eerder beschreven variabele
- datatype kolom uit databank

Syntax:

```
identifier    table.column_name%TYPE;  
identifier    other_variable%TYPE;
```

Vb.

v_getal	NUMBER(4,1);
v_getal2	v_getal%TYPE;
v_mndsal	employees.salary%TYPE;

→ dit best doen voor variabelen die hun waarde uit de database krijgen (zie later)

→ enkel het datatype wordt overgenomen, geen default-waarde

Declaratie: default-waarde

- zonder DEFAULT-waarde: NULL
- bij beschrijving NOT NULL of CONSTANT → DEFAULT-waarde verplicht

Vb

v_account	NUMBER(11)	NOT NULL	:= 1200000;
v_bonus	NUMBER(2)	DEFAULT 0;	
v_naam1	VARCHAR2(20)	DEFAULT 'X';	
v_naam2	v_naam1%TYPE	DEFAULT 'Y';	
v_vandaag	DATE	DEFAULT SYSDATE;	
v_gisteren	DATE	DEFAULT SYSDATE - 1;	
v_max	CONSTANT NUMBER(4)	:= 5000;	

Operatoren

- rekenkundig

** * / + -

- alfanumeriek

||

- vergelijking

= != <> < <= > >=

IS NULL

LIKE

BETWEEN

IN

- logisch

AND OR NOT

Oefening 1

Voorbeeld: functie met parameters

```
CREATE OR REPLACE FUNCTION fulldate  
(p_date IN DATE)  
    RETURN VARCHAR2  
  
AS  
  
BEGIN  
    RETURN TO_CHAR(p_date, 'fmDay, dd month yyyy');  
  
END;
```

Parameterlijst

(p_tekst
p_sal
...)

naam

p_...

varchar2,
employees.salary%TYPE,

datatype

(geen lengte!!)

Meerdere parameters zijn gescheiden door een komma

Functie gebruiken

- Vanuit een andere functie:

in het BEGIN-blok: `v_tekst := fulldate(v_datum);`

- Vanuit een SQL-instructie:

SQL> `SELECT fulldate(hire_date) FROM employees;`

Oefening 2

Oefening 3a

Voorwaardelijke uitvoering: IF

Syntax:

```
IF condition THEN
    statements;
[ELSIF condition THEN
    statements;]
[ELSE
    statements;]
END IF;
```

Voorwaardelijke uitvoering

Voorbeeld

```
IF v_leeftijd > 60 THEN
    v_categorie := 'senior';
ELSE
    v_categorie := 'middelbaar';
END IF;
```

Voorwaardelijke uitvoering: geneste IF

Voorbeeld

```
IF v_leeftijd > 60 THEN
    v_categorie := 'senior';
ELSE
    IF v_leeftijd > 35 THEN
        v_categorie := 'middelbaar';
    ELSE
        v_categorie := 'jong';
    END IF;
END IF;
```

Voorwaardelijke uitvoering: ELSIF

Voorbeeld

```
IF v_leeftijd > 60 THEN
    v_categorie := 'senior';
ELSIF v_leeftijd > 35 THEN
    v_categorie := 'middelbaar';
ELSE
    v_categorie := 'jong';
END IF;
```


Voorwaardelijke uitvoering: AND en OR

Voorbeeld1

```
IF v_leeftijd > 60 AND status = 'niet werkend' THEN  
    v_categorie := 'gepensioneerde senior';  
END IF;
```

Voorbeeld2

```
IF v_leeftijd < 18 OR status = 'student' THEN  
    v_belastingen := 0;  
END IF;
```

Oefening 3b

Oefening 4

Oefening 5

SELECT Statements in PL/SQL

- Gegevens uit de databank ophalen met een `SELECT` statement.
- Syntax:

```
SELECT  select_list
INTO    {variable_name[, variable_name]...
        | record_name}
FROM    table
[WHERE  condition];
```

- De INTO-clause is verplicht!
- Een query MOET 1 rij ophalen! → WHERE-clausule

SELECT Statements in PL/SQL: voorbeeld

```
CREATE OR REPLACE FUNCTION get_aantal_dienstjaren
(p_emp_id          employees.employee_id%TYPE)
RETURN NUMBER
AS
    v_hire_date          employees.hire_date%TYPE;
    v_aantal_jaren_dienst NUMBER(2);
BEGIN
    SELECT hire_date
    INTO v_hire_date
    FROM employees
    WHERE employee_id = p_emp_id;
    v_aantal_jaren_dienst := TRUNC(MONTHS_BETWEEN(sysdate, v_hire_date)/12);
    RETURN v_aantal_jaren_dienst;
END;
```

SELECT Statements in PL/SQL: opmerkingen

- volledige syntax van select statement kan gebruikt worden, incl. WHERE, GROUP BY en HAVING
- meer dan 1 variabele kan gevuld worden
aantal expr. in SELECT = aantal variabelen

Voorbeeld

```
SELECT department_name, SUM(salary)
INTO v_dep_name, v_som
FROM employees JOIN departments USING(department_id)
WHERE department_id = 80;
```

Programming Guidelines

- Maak je code leesbaarder en beter onderhoudbaar:
 - Kies duidelijke namen voor je variabelen
 - Zorg voor een duidelijke inspringing

```
BEGIN
  IF x = 0 THEN
    y:=1;
  END IF;
END;
/
```

```
...
AS
  v_deptno          NUMBER(4);
  v_location_id     NUMBER(4);
BEGIN
  SELECT  department_id,
          location_id
  INTO    v_deptno,
          v_location_id
  FROM    departments
  WHERE   department_name
          = 'Sales';

...
END;
/
```

Functies verwijderen

- Syntax:

```
DROP FUNCTION function_name
```

- Voorbeeld: `DROP FUNCTION get_jaarsal;`
 - Alle privileges betreffende de functie worden mee verwijderd.
 - De `CREATE OR REPLACE` syntax is equivalent aan het verwijderen en opnieuw creëren van de functie. Toegekende privileges i.v.m. de functie blijven bestaan als deze syntax gebruikt wordt.

Opvragen kenmerken (data dictionary)

`desc naam_functie`

→ Overzicht van invoer- en uitvoerparameters van de functie

Voorbeeld

```
SQL> desc netto
FUNCTION netto RETURNS VARCHAR2
Argument Name          Type                      In/Out Default?
-----
P_BRUTO                 NUMBER(8,2)              IN
```


Opvragen kenmerken (data dictionary)

Alle informatie over bestaande PL/SQL functies is bewaard in de databank. Je kan hiervoor gebruik maken van volgende Oracle data dictionary views:

- `USER_OBJECTS`: deze view bevat informatie over ALLE databankobjecten van de eigen user, dus alle zelf-gecreëerde tabellen, indexen, sequences, functies,...
- `USER_SOURCE`: hierin zit de code van bepaalde objecten

Opvragen kenmerken (data dictionary)

USER_OBJECTS

belangrijkste kolommen zijn object_name, object_type, created, ...

Voorbeeld om te kijken welke functies aanwezig zijn:

```
SELECT object_name
FROM   user_objects
WHERE  object_type = 'FUNCTION';
```

Opvragen kenmerken (data dictionary)

USER_SOURCE

belangrijkste kolommen zijn name, type, line, text

Voorbeeld om de code van een bestaande functie te bekijken:

```
SELECT text
FROM   user_source
WHERE  name = 'GET_JAARSAL';
```

Oefeningen

PL/SQL H3

Procedures



Wat is een procedure?

- object (bestaat uit SQL-statements en PL/SQL-constructies) met een naam – ook subprogramma genoemd
- wordt bewaard in de DB
- code op 1 plaats definiëren en op meerdere plaatsen gebruiken
- voert één of meerdere acties uit
- kan aangeroepen(called) worden met 1 of meerdere parameters
- slechts 1 aanroep voor meerdere acties waardoor betere performance

Syntax voor de creatie van een procedure

```
CREATE [OR REPLACE] PROCEDURE procedure_name  
[(parameter1 [mode] datatype1,  
  parameter2 [mode] datatype2, ...)]  
IS|AS  
  [local_variable_declarations; ...]  
BEGIN  
  -- actions;  
END [procedure_name];
```

Voorbeeld: procedure zonder parameters

```
CREATE OR REPLACE PROCEDURE show_emp
```

```
IS
```

```
    v_emp            employees.employee_id%type := '100';
```

```
    v_voornaam       employees.first_name%TYPE;
```

```
    v_naam           employees.last_name%TYPE;
```

```
BEGIN
```

```
    SELECT first_name, last_name
```

```
    INTO v_voornaam, v_naam
```

```
    FROM employees
```

```
    WHERE employee_id = v_emp;
```

```
    DBMS_OUTPUT.PUT_LINE(v_voornaam || ' ' || v_naam);
```

```
END show_emp;
```

Let op: als employee_id 100 niet bestaat in de tabel employees dan zal de procedure afsluiten met de foutmelding: “no data found”

Afdrukken

- Gebruik maken van ingebouwde package DBMS_OUTPUT met mogelijke procedures o.a. PUT_LINE
- Tussen ronde haakjes de af te drukken lijn als 1 geheel meegeven
- Vb. DBMS_OUTPUT.PUT_LINE('Je kan tekst en variabelen samenvoegen met concatenatie-teken' || ' alle tekst moet tussen single quotes. Dit moet niet voor variabelen.' || v_country_name);
- in SQL*Plus SET SERVEROUTPUT ON opnemen → best in LOGIN.SQL
- LOGIN.SQL is een script dat telkens bij het opstarten van SQL*Plus wordt uitgevoerd en waar bepaalde settings kunnen opgenomen worden

/ → creatie procedure

- de broncode wordt in ieder geval in de data dictionary opgeslagen
- als foutloze code: gecompileerde versie → databank
- als code met fouten:
Melding: ``created with compilation errors'``.

Hoe fouten opvragen: `show errors`

Voorbeeld: procedure zonder parameters

```
CREATE OR REPLACE PROCEDURE add_ctr  
IS  
    v_country_id          countries.country_id%type := 'FR';  
    v_country_name        countries.country_name%type := 'France';  
    v_region_id           countries.region_id%type := 1;  
BEGIN  
    INSERT INTO countries  
    VALUES (v_country_id,v_country_name,v_region_id);  
    DBMS_OUTPUT.PUT_LINE('Er werden ' || SQL%ROWCOUNT || ' rijen  
    toegevoegd in de tabel COUNTRIES');  
END add_ctr;  
/
```

PL/SQL Syntax

- Elk DML-statement(INSERT, UPDATE, DELETE) kan zonder aanpassingen aan syntax opgenomen worden in PL/SQL
- Let op: een **DML-statement** in PL/SQL geeft geen fout als er 0 of meerdere rijen bewerkt worden
- Een **SELECT-statement** in PL/SQL dat 0 resultaatrijen geeft zal leiden tot de foutmelding 'NO DATA FOUND'
- Een **SELECT-statement** in PL/SQL dat meerdere resultaatrijen geeft zal leiden tot de foutmelding 'EXACT FETCH RETURNS MORE THAN REQUESTED NUMBER OF ROWS'

Impliciete cursor

Info omtrent het laatste SQL-statement staat in cursor **SQL**

- Cursor is een pointer naar een stukje geheugen
- Expliciete cursor: aangemaakt door gebruiker
- Impliciete cursor: aangemaakt door de Oracle Server o.a. **SQL**
- Toegang tot de info in de cursor kan via cursor attributen:

SQL%FOUND	Boolean attribute that evaluates to TRUE if the most recent SQL statement returned at least one row.
SQL%NOTFOUND	Boolean attribute that evaluates to TRUE if the most recent SQL statement did not return even one row.
SQL%ROWCOUNT	An integer value that represents number of rows affected by the most recent SQL statement.

Procedure oproepen

- Vanuit een andere procedure:

in het BEGIN-blok: **ADD_CTRY;**

- Vanuit SQL*Plus:

SQL> execute add_ctry

OF exec add_ctry

Oefening 1

Oefening 2

Voorbeeld: procedure met IN parameter(s)

```
CREATE OR REPLACE PROCEDURE del_etry
    (p_country_id      IN      countries.country_id%type)
IS
BEGIN
    DELETE FROM countries
    WHERE country_id = p_country_id;

    DBMS_OUTPUT.PUT_LINE('Er werden ' || SQL%ROWCOUNT ||
                          'rijen verwijderd uit de tabel COUNTRIES');

END del_etry;
/
```


PL/SQL Syntax - parameters

- Parameterlijst: zie hoofdstuk Functies
- IN-parameter komt binnen in de procedure (called program, subprogramma) en wordt meegegeven vanuit een andere procedure (calling program) of vanuit een aanroep in SQL*Plus
- De parameter(s) in het called program worden FORMAL-parameters genoemd
- De parameter(s) in het calling program worden ACTUAL-parameters genoemd
- Sql> `DESC[RIBE] del_ctype` geeft informatie over de parameters van de procedure del_ctype

Procedure oproepen

- Vanuit een andere procedure:

in het BEGIN-blok:

```
del_ctry(v_country_id) ;
```

```
OF del_ctry('AR')
```

- Vanuit SQL*Plus:

```
SQL> exec del_ctry('AR')
```

```
OF exec del_ctry('&landid')
```

Oefening 3

Voorbeeld: procedure met IN en OUT parameter(s)

```
CREATE OR REPLACE PROCEDURE raise_salary_dept
    (p_dept_name      IN      departments.department_name%type
    ,p_percent         IN      number
    ,p_count_emp       OUT     number)
AS
    v_dept_id departments.department_id%type;
BEGIN
    SELECT department_id INTO v_dept_id
    FROM departments
    WHERE department_name = p_dept_name;
    UPDATE employees
    SET salary = salary * (1 + p_percent/100)
    WHERE department_id = v_dept_id;
    p_count_emp := SQL%rowcount;
END raise_salary_dept;
```

Voorbeeld: Andere mogelijke oplossing

```
CREATE OR REPLACE PROCEDURE raise_salary_2_dept
    (p_dept_name    IN      departments.department_name%type
    ,p_percent      IN      number
    ,p_count_emp    OUT     number)
AS
BEGIN
    UPDATE employees
    SET salary = salary * (1 + p_percent/100)
    WHERE department_id = (SELECT department_id FROM departments
                           WHERE department_name = p_dept_name);

    p_count_emp := SQL%rowcount;
END raise_salary_2_dept;
/
```

Procedure met IN-en OUT parameter(s) oproepen

Een procedure kan worden opgeroepen

- vanuit een andere procedure
- via een anoniem block
- aan de SQL-prompt → bind-variable nodig

Oproeping vanuit een andere procedure

.....

AS

 v_aantal_emp number(3);

BEGIN

 raise_salary_dept('Administration', 10, v_aantal_emp);

 DBMS_OUTPUT.PUT_LINE(v_aantal_emp);

END;

- Vanuit deze procedure wordt de naam van het department nl. Administration en het percentage nl. 10 meegegeven aan het called program nl. raise_salary_dept
- Na het uitvoeren zal het aantal employees in het department Administration met een loonsverhoging van 10% worden afgedrukt

Oproeping via een anoniem block

```
DECLARE
    v_aantal_emp    number(3);
BEGIN
    raise_salary_dept('Administration', 10, v_aantal_emp);
    DBMS_OUTPUT.PUT_LINE(v_aantal_emp);
END;
/
```


Oproeping via de SQL-prompt → bind-variable

- Bind variable
 1. deze variabele wordt gecreëerd in de werkomgeving en kan gebruikt worden in SQL statements en PL/SQL blocks
 2. syntax aan SQL-prompt: VARIABLE b_test varchar2(2)
 3. gebruikt als volgt :b_test
- SQL> VARIABLE b_aantal_emp number
SQL> exec raise_salary_dept('Administration', 10, :b_aantal_emp)
- Om afdruk van bind variable te zien voeg je de volgende setting toe in login.sql: SET AUTOPRINT ON

```
SQL> exec raise_salary_dept('Administration',10,:b_aantal_emp)

PL/SQL procedure successfully completed.

B_AANTAL_EMP          1
```

Oefening

Probeer de procedure `raise_salary_2_dept` uit te voeren en te gebruiken (Voorbeeld: Andere mogelijke oplossing)

Positional vs. Named notation

- Voorbeeld positional notation
`exec raise_salary_dept('Administration', 10, :b_aantal_emp)`

Volgorde van parameters moet exact dezelfde zijn als in de procedure

- Voorbeeld named notation
`exec raise_salary_dept(p_percent =>10,p_dept_name
=>'Administration',p_count_emp =>:b_aantal_emp)`

Volgorde is niet langer belangrijk maar de parameter-namen moeten dan wel gekend zijn

Oefening 4

Voorbeeld: procedure met eenvoudige LOOP

```
CREATE OR REPLACE PROCEDURE print_dept_loop
AS
    v_count          number(3) := 0;
    v_dept_id        departments.department_id%type;
    v_dept_name      departments.department_name%type;
    v_man_id         departments.manager_id%type;
BEGIN
    LOOP
        v_count := v_count + 10;
        SELECT department_id, department_name, manager_id
        INTO v_dept_id, v_dept_name, v_man_id
        FROM departments
        WHERE department_id = v_count;
        DBMS_OUTPUT.PUT_LINE(v_dept_id||'   '||v_dept_name||'   '||v_man_id);
        EXIT WHEN v_count >=100;
    END LOOP;
END print_dept_loop;

/
```

(herhaal tot ...)

Voorbeeld: procedure met WHILE LOOP

```
CREATE OR REPLACE PROCEDURE print_dept_while  
AS
```

```
    v_count          number(3) := 10;  
    v_dept_id        departments.department_id%type;  
    v_dept_name      departments.department_name%type;  
    v_man_id         departments.manager_id%type;
```

```
BEGIN
```

```
    WHILE v_count <= 100 LOOP  
        SELECT department_id, department_name, manager_id  
        INTO v_dept_id, v_dept_name, v_man_id  
        FROM departments  
        WHERE department_id = v_count;  
        DBMS_OUTPUT.PUT_LINE(v_dept_id||'   '||v_dept_name||'   '||v_man_id);  
        v_count := v_count + 10;  
    END LOOP;
```

```
END print_dept_while;
```

(zolang ...)

BIG DATA - PL/SQL - H3 - Procedures

Voorbeeld: procedure met FOR LOOP

```
CREATE OR REPLACE PROCEDURE print_dept_for
AS
```

```
    v_dept_id      departments.department_id%type;
    v_dept_name     departments.department_name%type;
    v_man_id        departments.manager_id%type;
```

```
BEGIN
```

```
    FOR i IN 1..10 LOOP
```

```
        SELECT department_id, department_name, manager_id
```

```
        INTO v_dept_id, v_dept_name, v_man_id
```

```
        FROM departments
```

```
        WHERE department_id = i*10;
```

```
        DBMS_OUTPUT.PUT_LINE(v_dept_id||'   '||v_dept_name||'   '||v_man_id);
```

```
    END LOOP;
```

```
END print_dept_for;
```

```
/
```

(zelftellende lus)

OPM : teller i wordt impliciet gedeclareerd en kan enkel in lus worden gebruikt – kan wel toegewezen worden aan een variabele en deze is bruikbaar buiten de lus

Oefening 5

Oefening 6

Speciale variant van de FOR-loop – Cursor Loop

```
CREATE OR REPLACE PROCEDURE print_dept_cursorloop
AS
BEGIN
    FOR rec IN (SELECT department_id, department_name, manager_id
                FROM departments
                WHERE department_id between 10 and 100)
    LOOP
        DBMS_OUTPUT.PUT_LINE(rec.department_id || ' '
                              || rec.department_name || ' ' || rec.manager_id);
    END LOOP;
END print_dept_cursorloop;
```

Speciale variant van de FOR-loop – Cursor Loop

- Alle rijen en kolommen bekomen door het uitvoeren van de subquery worden in een expliciete cursor bijgehouden.
- Deze cursor heeft geen naam en daarom kan er ook geen gebruik gemaakt worden van cursorattributen
=> `rec%rowcount` kan NIET GEBRUIKT worden
- De FOR-loop zal rij per rij verwerken
- In de LOOP kan er verwezen worden naar een specifiek attribuut via `rec.department_name` – het gaat hier dan over de inhoud van het attribuut `department_name` in de rij die op dat moment door de loop verwerkt wordt

Oefening 7

Oefening 8

Procedures verwijderen

- Syntax:

```
DROP PROCEDURE procedure_name
```

- Voorbeeld: `DROP PROCEDURE raise_salary_dept;`
 - Alle privileges betreffende de procedure worden mee verwijderd.
 - De `CREATE OR REPLACE` syntax is equivalent aan het verwijderen en opnieuw creëren van de procedure. Toegekende privileges i.v.m. de procedure blijven bestaan als deze syntax gebruikt wordt.

Opvragen kenmerken (data dictionary)

Alle informatie over bestaande PL/SQL procedures is bewaard in de databank. Je kan hiervoor gebruik maken van volgende Oracle data dictionary views:

- **USER_OBJECTS**: deze view bevat informatie over ALLE databankobjecten van de eigen user, dus alle zelf-gecreëerde tabellen, indexen, sequences, functies, procedures,....
- **USER_SOURCE**: hierin zit de code van bepaalde objecten

Opvragen kenmerken (data dictionary)

USER_OBJECTS

belangrijkste kolommen zijn object_name, object_type, created, ...

Voorbeeld om te kijken welke procedures aanwezig zijn:

```
SELECT object_name  
FROM   user_objects  
WHERE  object_type = 'PROCEDURE';
```

Opvragen kenmerken (data dictionary)

USER_SOURCE

belangrijkste kolommen zijn name, type, line, text

Voorbeeld om de code van een bestaande procedure te bekijken:

```
SELECT text
FROM   user_source
WHERE  name = 'RAISE_SALARY_DEPT' ;
```

Oefening 9

Oefening 10

PL/SQL H4


Database Triggers

HOGESCHOOL 



Inleiding Database Triggers

- PL/SQL-code geassocieerd aan een tabel/view
- wordt als object opgeslagen in de DB
- wordt automatisch uitgevoerd (kan niet opgeroepen worden)



Inleiding Database Triggers

- inhoud van de tabel wijzigt (insert/update/delete)
  trigger wordt automatisch uitgevoerd

!!!!!! het maakt niet uit vanuit welke omgeving of door wie de gegevens worden gewijzigd!!!!!!

Syntax voor de creatie van een database trigger

CREATE OR REPLACE TRIGGER *triggernaam*

{BEFORE / AFTER}

→ ***timing***

{DELETE / INSERT / UPDATE [OF *kolom* [,*kolom*]] } [OR...]

→ ***event***

ON *tabelnaam*

[FOR EACH ROW [WHEN (*voorwaarde*)]]

→ ***frequency***

[DECLARE]

BEGIN

*/*uitvoerbare commando's*/*

[EXCEPTION]

END [*triggernaam*];

Syntax tabel triggers


- **timing**
 - BEFORE the event
 - AFTER the event
 - ☐ BEFORE
 - als de trigger moet controleren of een actie toegestaan is of niet - voorkomt onnodige rollbacks
 - als je zeker wil zijn dat deze trigger altijd afgaat (deze worden eerst uitgevoerd)
 - ☐ AFTER
 - als het triggerende commando zeker moet worden uitgevoerd (vooral bij row triggers)
 - wordt pas uitgevoerd nadat alle constraints op de tabel gecontroleerd zijn
- **event**
 - INSERT – UPDATE - DELETE
- **frequency**
 - STATEMENT (default)
 - ROW
 - = aantal keer trigger-body uitgevoerd wordt
 - ☐ statement:
 - 1x per statement / instructie
 - onafhankelijk van het aantal beïnvloede rijen
 - ☐ row:
 - 1x per rij die beïnvloed wordt door het event

Voorbeeld: Statement Trigger

```
CREATE OR REPLACE TRIGGER bds_emp  
  BEFORE DELETE  
  ON employees  
BEGIN  
  IF USER != 'JAN' THEN  
    RAISE_APPLICATION_ERROR(-20000,  
      'u heeft geen rechten voor deze actie');  
  END IF;  
END;  
/
```

creatie als object in de databank

Foutmelding via RAISE_APPLICATION_ERROR

- RAISE_APPLICATION_ERROR(-20000, 'u heeft geen rechten voor deze actie')


foutcode foutmelding(moet string zijn)
- Een SQL-commando waarmee een fout gecreëerd/geraised wordt en op het scherm wordt afgedrukt (bruikbaar in elke applicatie)
 - het programma/trigger wordt afgebroken
 - automatisch ROLLBACK voor het triggerende DML-statement (hier delete)
- Foutcode moet liggen tussen -20000 en -20999 (user-defined)

Naamgeving triggers

De naamgeving geeft het soort trigger weer

vb: TRIGGER bds_emp

- b = before
de trigger gaat af vóór het DML-statement wordt uitgevoerd
- d = delete
de trigger gaat af bij een delete-statement
- s = statement trigger
deze trigger gaat slechts 1x af per DML-statement, ongeacht hoeveel rijen door het DML-commando worden bewerkt

/ → creatie trigger

- Via R(un) of / indien code in buffer of via start [naam sql-bestand]
- de broncode wordt in ieder geval in de data dictionary opgeslagen
- als foutloze code: gecompileerde versie → databank
- als code met fouten:
Melding: ``created with compilation errors'``.

Hoe fouten opvragen?: `→ show errors`

Trigger gebruiken?

- De tabel trigger gaat automatisch af bij het uitvoeren van een DML-statement op een specifieke tabel waarvoor een trigger gecompileerd is
- Er kunnen meerdere triggers op 1 tabel gecreëerd worden (volgorde van uitvoeren: zie later)

Oefening 1

Voorbeeld: Statement Trigger uitgebreid

```
CREATE OR REPLACE TRIGGER bdus_emp
  before delete or update of salary
  ON employees
BEGIN
  IF USER != 'JAN' THEN
    IF DELETING THEN
      RAISE_APPLICATION_ERROR(-20000, 'u heeft geen verwijderrechten');
    ELSE
      RAISE_APPLICATION_ERROR(-20000, 'u heeft geen rechten om het salary te wijzigen');
    END IF;
  END IF;
END;
/
```

Functies INSERTING – DELETING - UPDATING

- Te gebruiken indien er meer dan 1 triggerend event is op 1 tabel
- Deze functies geven een boolean terug
- Bij UPDATING kan ook een parameter meegegeven worden
vb. IF updating('salary') THEN

Oefening 2

Oefening 3

Voorbeeld: Row Trigger

```
CREATE OR REPLACE TRIGGER aur_emp_salary
  AFTER UPDATE OF salary
  ON employees
  FOR EACH ROW
BEGIN
  IF (:NEW.salary - :OLD.salary > 0.1* :OLD.salary) THEN
    RAISE_APPLICATION_ERROR(-20000, 'salary te veel
                                   verhoogd' ) ;
  END IF;
END;
/
```

Row triggers

- bovenaan de trigger definitie: FOR EACH ROW
- gaat per te bewerken rij af
 - Vb een delete-commando verwijdt 10 rijen
 - statement-trigger gaat 1 keer af
 - rij-trigger gaat 10 keer af
 - Vb een delete-commando verwijdt 0 rijen
 - statement-trigger gaat 1 keer af
 - rij-trigger gaat niet af

Row triggers

- mogelijk oude en nieuwe kolomwaarden op te vragen
 - :NEW.kolomnaam
 - :OLD.kolomnaam

vb. :NEW.salary bevat de nieuwe waarde voor salary na uitvoering van een INSERT of UPDATE
Let op: bij DELETE is deze variabele leeg

:OLD.salary bevat de oude waarde voor salary vóór uitvoering van een UPDATE of DELETE
Let op: bij INSERT is deze variabele leeg

Enkel bij UPDATE bevatten beiden een waarde

Voorbeeld: Row Trigger uitbreiding (WHEN)

```
CREATE OR REPLACE TRIGGER aur_emp_sal2
  AFTER UPDATE OF salary
  ON employees
  FOR EACH ROW
  WHEN (OLD.job_id != 'AD_PRES')
BEGIN
  IF (:NEW.salary - :OLD.salary > 0.1* :OLD.salary) THEN
    RAISE_APPLICATION_ERROR(-20000, 'salary te veel
                                   verhoogd');
  END IF;
END;
/
```

Enkel mogelijk voor row triggers!

LET OP: geen ':' bij 'OLD'

Oefening 4

Oefening 5

Volgorde van uitvoering triggers (automatisch)

Indien meerdere triggers op 1 DML-statement

1. Alle BEFORE STATEMENT triggers
2. Voor elke rij uit de ROW triggers
 - a. Alle BEFORE ROW triggers voor die rij
 - b. Triggerende DML-statement + integrity constraints checken voor die rij
 - c. Alle AFTER ROW triggers voor die rij
3. Alle AFTER STATEMENT triggers

Trigger keuze

- Gebruik een **row trigger** als de inhoud van de kolommen nodig is
- Gebruik een **before statement trigger** als de trigger MOET afgaan
- Gebruik eerder een **before statement-trigger** dan een after statement trigger
(vooral bij controle of een actie toegestaan is, dit voorkomt rollbacks)
- Gebruik liever een **after row trigger** dan een before row trigger.
Oracle controleert dan eerst de constraints.
- Gebruik een **before row trigger** als de inhoud van een kolom in de trigger gewijzigd wordt

Beperkingen van Database Triggers

Commando's COMMIT en ROLLBACK zijn niet toegelaten in triggers

Oefening 6

Beheer van triggers

- **Welke triggers bestaan?**

```
SQL>SELECT object_name, created, status  
      FROM user_objects  
      WHERE object_type = 'TRIGGER';
```

- **Broncode opvragen:**

```
SQL> SELECT line, text  
      FROM user_source  
      WHERE name = 'AUR_EMP_SALARY';
```


Beheer van triggers – tabel **USER_TRIGGERS**

- SQL> SELECT trigger_type, trigger_body
FROM **user_triggers**
WHERE trigger_name = 'AUR_EMP_SALARY';
- SQL> SELECT trigger_name, trigger_type,
triggering_event, table_name, status
FROM **user_triggers**;

Beheer van triggers

- **Verwijderen**

DROP TRIGGER triggernaam

- **Activeren/deactiveren van 1 bepaalde trigger**

ALTER TRIGGER triggernaam ENABLE

ALTER TRIGGER triggernaam DISABLE

- **Activeren/deactiveren van alle triggers van 1 bepaalde tabel**


ALTER TABLE tabelnaam ENABLE ALL TRIGGERS

ALTER TABLE tabelnaam DISABLE ALL TRIGGERS

Beheer van triggers

- een trigger die **invalid** geworden is, kan opnieuw worden gecompileerd:

```
ALTER TRIGGER triggernaam COMPILE
```



Hoe kan een trigger invalid worden?

wanneer bvb een wijziging van de structuur van de gerefereerde tabel optreedt

Oefening 7